



קובץ הכנה ניסוי מעבדה מס' 2

## **Tutorial 2.2 – Project Organization**

מעבדת מיקרומחשבים – המחלקה להנדסת חשמל ומחשבים

מס' קורס - 361.1.3353

כתיבה ועריכה: חנן ריבוא

מהדורה 1 – שנה"ל תשע"ו

בתרגול אסמבלי זה, נלמד איך ליצור מודולים של ספרייה ולכתוב פרויקט גדול בעזרת מספר קובצי מקור (לא רק בקובץ מקור אחד, כפי שעשיתם עד כה).

## A. הקדמה:

1. במידה ואנו עובדים על פרויקט בהיקף קוד גדול (המכיל הרבה רוטינות וקטעי קוד רבים) נרצה לחלק את הפרויקט למספר קובצי מקור כדי להימנע מתוכנית אחת הכתובה לאורך קובץ מקור אחד (דבר הגורם לסרבול ולבלבול רב).
2. קבצים אלה אשר מכילים מודולים של ספרייה, נקראים object files כלומר: קבצים אלו הם חלק מהפרויקט אבל עדיין לא יעברו קישור (LINK) אלא אם כן יקראו להם דרך התוכנית.
3. הגדרת ספרייה = אוסף של רוטינות ב-object file אחד.
4. שימוש במודלי ספרייה מומלץ מאוד ליכולת תחזוקת קוד האסמבלי.
5. ישנן שתי גישות לעבודה עם ספריות:

**גישה 1:** נחלק את כתיבת הפרויקט על פני מספר קובצי מקור.

קובץ מקור אחד עבור התוכנית הראשית וקובצי מקור נוספים עבור הספריות.

בסביבת העבודה (IAR Workspace) כל קבצי מקור המיועדים לספרייה יהיו גלויים לפנינו.

יתרון: כל קבצי הספרייה גלויים לפנינו נוכל לעדכןם מידי (נוחות טכנית).


חסרון: בכל פעם שנעשה מחדש Build לתוכנית שלנו (גם לאחר ששינינו משהו קטן בתוכנית הראשית) השלבים עד להמרה לקוד מכונה מתבצעים במלואם לכל הקבצים (כולל קבצי ספרייה שלא עודכנו כלל) דבר הגורם לפעולות מיותרות. עבור פרויקט גדול זמן זה מורגש.

**גישה 2:** נעבוד עם קובץ מקור אחד ונהפוך את כל קובצי המקור שהיו גלויים (בגישה 1) לספריות שאינן גלויים. היתרונות והחסרונות דואליים לאלו שבגישה 1.

6. הצגת מספרי השורות על גבי מסך ה-Editor:

לצורך נוחות, ניתן להציג את מספרי השורות על גבי מסך ה-Editor, לפי השלבים הבאים:

עמוד עם העכבר על מרכז המסך, לחצן ימני ולחיצה על options ולאחר מכן בצע את המוצג בתצלום הבא.



**B. דוגמה 1 - קוד אסמבלי מחולק לשני קבצים:**

הקוד מכיל שני קבצים (Main.s43 , Functions.s43).

התוכנית מבצעת דרך P9 הדלקה לד בודדת בכל פעם, של לד אחר לד משמאל לימין בצורה מחזורית. השהיה בין הדלקת הלדים של 0.2sec , השהיה בין הדלקת הלד השמיני לראשון (בין מחזור לדים) היא 1sec.

**קובץ 1 - Main.s43**

```

1  #include <msp430xG46x.h>
2  ;
3      NAME      Main
4      PUBLIC    Main
5      EXTERN    P9Config, SkipLedSerial
6      EXTERN    Delay100us, Delay1sec
7
8      RSEG      CSTACK
9      RSEG      CODE
10 Main          mov.w    #SFE(CSTACK), SP
11 StopWDT       mov.w    #WDTPW+WDTHOLD, &WDTCTL
12              call     #P9Config
13
14 L              call     #SkipLedSerial
15              call     #Delay1sec
16              jmp      L
17              nop
18              ; for debugging
19 ;*****
19      COMMON    INTVEC    ;Interrupt Vectors-Begins a common segment INTVEC
20 ;*****
21      ORG      RESET_VECTOR    ; MSP430 RESET Vector
22      DW      Main
23      END

```

**✓ תיאור מבנה קובץ 1 - Main.s43 (קובץ תוכנית ראשית).**

- שורה 3: הגדרת מודול ע"י **NAME** ונתינת שם Main למודול.
- שורה 4: הרשאה שמודול זה יהיה "ציבורי" ע"י **PUBLIC** וניתן לשימוש עבור מודולים אחרים.
- שורות 5-6: הצהרה על שימוש בתוויות (labels) המוגדרות מחוץ למודול זה, ע"י הוראת **EXTERN**. במקרה זה מדובר על רוטינות הנמצאות בקובץ **Functions.s43**.
- שורות 8-23: הגדרת קוד של התוכנית הראשית.

**✓ תיאור מבנה קובץ 2 - Functions.s43 (קובץ ספרייה).**

- שורות 5-10: הגדרת מקרו, תמיד תירשם לאחר הוראות **#include** ו- **#define**.
- שורות 12-14: שימוש בהוראת **#define** לצורך ביצוע גנרי של הקוד.
- שורה 16: פתיח למודול בשם PortLeds ע"י **NAME**.
- שורה 17: הרשאת **PUBLIC** לשתי הרוטינות P9Config , SkipLedSerial המוגדרות בתוך המודול.
- שורה 18: הוראת **EXTERN** לשימוש ברוטינות Delay100us, Delay1sec המוגדרות מחוץ למודול זה.
- שורות 19-36: הגדרת קוד המודול.
- שורה 8: סיום המודול ע"י הוראת **ENDMOD**.

- שורות 39-54: הגדרת מודול המכיל שתי רutiנות Delay100us, Delay1sec לפי הכללים לעיל.
- שורה 56: הוראת סיום הקובץ.

## קובץ 2 - Functions.s43

```

1  #include <msp430xG46x.h>
2  ;-----
3  ;          MACRO Defenition
4  ;-----
5  delaymac  MACRO    time
6             local   L3
7             mov.w   time,R15
8  L3        dec.w    R15
9             jnz     L3
10            ENDM
11 ;-----
12 #define    LedPortSel    &P9SEL
13 #define    LedPortDir    &P9DIR
14 #define    LedPort       &P9OUT
15 //=====
16            MODULE        PortLeds
17            PUBLIC        P9Config, SkipLedSerial
18            EXTERN        Delay100us, Delay1sec
19            RSEG          CODE
20
21 P9Config    bic.b        #0xff,LedPortSel
22            bis.b        #0xff,LedPortDir
23            clr.b        LedPort
24            ret
25
26
27 SkipLedSerial  mov.b        #0x01,R14
28 Subloop      mov.b        R14,&P9OUT
29            call         #Delay100us    ;delay using routine
30            delaymac     #52429        ;delay using Macro
31            rla.b        R14
32            tst.b        R14
33            jz          finish
34            jmp         Subloop
35 finish      ret
36
37            ENDMOD
38 //=====
39            MODULE        Delay
40            PUBLIC        Delay100us, Delay1sec
41            RSEG          CODE
42
43 Delay100us   mov.w        #52429,R15
44 L1          dec.w        R15
45            jnz         L1
46            ret
47
48 Delay1sec    mov.w        #10,R13
49 L2          call         #Delay100us
50            dec.w        R13
51            jnz         L2
52            ret
53
54            ENDMOD
55 //=====
56            END

```

הגדרת פונקציית ה-MACRO צריכה להופיע בראש קובץ המקור וצריכה להופיע בראש כל קובץ מקור נוסף, במידה וקוראים לה ממנו.

**C. דגשים חשובים:**

- 1) בכל קובץ שממנו נרצה **להפעיל** פונקציית MACRO, נצטרך **להגדיר** בראש הקובץ את פונקציית ה-MACRO. דבר זה מאפשר להפעיל את ה-MACRO מכל MODULE בקובץ.
- 2) ניתן להגדיר בתוך MODULE מספר רוטיות, אולם יש להצהיר עליהם כ- **PUBLIC**.
- 3) כדי שנוכל לקרוא לרוטיות המוגדרות ב-MODULE אחר, נצרך לרשום את שם הרוטינה שבה נרצה להשתמש בשורת ה- **EXTERN**.
- 4) רשימת שורת ה- #include בראש קובץ קוד, רלוונטי ל-MODULE הראשון בלבד בקובץ. לכן, אם נרצה לעבוד עם פורטים הקשורים לחומרה, נצטרך לבצע זאת רק ב-MODULE הראשון בלבד.

**D. סדר פעולות:****גישה 1:**

1. יוצרים פרויקט בדרך הרגילה וצירוף 2 הקבצים הנ"ל לפרויקט (קובץ הדרכה מופיע בתיקיית ניסוי מס' 1).
  2. לחיצה ימנית על העכבר ולבחור  
Project -> Options -> General Options -> Library configuration -> None (from the Library drop-down list)
- This means that a standard C/C++ library will not be linked – That is the default.
3. נותר לקמפל את הפרויקט (Project -> Compile).
  4. צרוב את הקוד לבקר.

**גישה 2:**

1. יוצרים פרויקט בדרך הרגילה ומצרפים רק את קובץ התוכנית הראשית (Main.s43) – יצירת workspace.
  2. בתיקיית הפרויקט שיצרנו, נפתח פרויקט חדש בשם LIB (שם כלשהו שתבחרו) ונצרך לפרויקט זה את קובץ הספרייה (Functions.s43).
- בפרויקט זה יש לבדוק את ההגדרות הבאות:

Project -> Options -> General Options ->	לשונית	בחירה
	Output	Output file: Library
	Library Configuration	Library: None

- לבסוף לחץ OK.
- לחץ על **Project -> Make** ובדוק שקובץ הספרייה **LIB.r43** נוצר בתיקיית `your_project\Debug\Exe`
  - 3. סגור את פרויקט LIB ופתח את הפרויקט הראשי וצרף אליו את הקובץ **LIB.r43**
  - 4. בפרויקט הראשי לחץ **Project -> Make** לצורך בניית הפרויקט.
  - 5. צרוב את הקוד לבקר.

**בניסויי המעבדה שלנו, מומלץ לעבוד לפי גישה 1**

**E. דוגמה 2 :**

הקוד מכיל שלושה קבצים (Main.s43 , Functions.s43 , Interrupts.s43).

לאחר ביצוע RESET, הקוד מבצע:

- הצגת ערך של שני משתנים אחד לאחר השני ולאחר מכן את סכומם.
- בלחיצה על P1.0 הלדים נדלקים זה אחר זה בלולאה אינסופית. לחיצה נוספת מקפיאה את מצב הלדים.

**(a)** קובץ Main.s43 – מכיל את קטע הקוד הראשי, ממנו נקרא לפונקציות. להלן סדר כתיבת הקוד:

1. #include <msp430xG46x.h>
2. **MACRO** Definitions
3. **NAME** Main
4. **PUBLIC** Main
5. All **EXTERN** routines labels
6. Continue with the regular code
7. **END**

**(b)** קובץ Functions.s43 – מכיל את קטע קוד למימוש הרוטינות. להלן סדר כתיבת הקוד:

1. #include <msp430xG46x.h> // **Only inside the first MODULE** if using HW PORT
2. **MACRO** Definitions
3. All **EXTERN** routines labels
4. **each MODULE** Routine Definitions:
  - ♦ **MODULE** label
  - ♦ **PUBLIC** label
  - ♦ **RSEG** CODE
  - ♦ Label
  - ♦ Write the routine body
  - ♦ Ret / Reti
  - ♦ **ENDMOD**
5. **END**

**(c)** קובץ Interrupts.s43 – מכיל את קטע קוד למימוש ה-ISR. סדר כתיבת הקוד, זהה לקובץ

Functions.s43

## Main.s43 - 1 קובץ

```

1  #include <msp430xG46x.h>
2  ;-----
3  ;          MACRO Defenition
4  ;-----
5  Delaymac      MACRO      time
6                  LOCAL    L
7                  mov       time,R15
8  L              dec        R15
9                  jnz       L
10                 ENDM
11 ;-----
12 ShowNum        MACRO      Num
13                 mov.b     Num,P9OUT
14                 ENDM
15 ;-----
16                 NAME      Main
17                 PUBLIC    Main
18                 EXTERN    P9Config, P1Config, PORT1_ISR
19                 EXTERN    Delay, Adder
20 ;-----
21                 ORG       0x1100
22 var1            DW        4
23 var2            DW        5
24 var3            DS        1
25 ;-----
26                 RSEG      CSTACK
27                 RSEG      CODE
28 Main            mov.w     #0x3100,SP
29                 mov.w     #WDTPW+WDTHOLD,&WDTCTL ; Stop WDT
30                 call      #P9Config
31                 call      #P1Config
32                 ShowNum    var1
33                 Delaymac   #65000 ;delay using MACRO
34                 ShowNum    var2
35                 call      #Delay ;delay using ROUTINE
36 ;-----
37                 push.w    var1
38                 push.w    var2
39                 call      #Adder
40                 incd      SP
41                 pop        var3
42                 ShowNum    var3
43 ;-----
44                 bis.w     #CPUOFF+GIE,SR ;LMP0
45                 nop       ; for break point
46 ;*****:
47                 COMMON    INTVEC ;Interrupt Vector segment name of INTVEC
48 ;*****:
49                 ORG       PORT1_VECTOR ;PORT1 Interrupt Vector
50                 DW        PORT1_ISR
51                 ORG       RESET_VECTOR ; MSP430 RESET Vector
52                 DW        Main
53                 END

```

## קובץ 2 - Functions.s43

```

1  #include <msp430xG46x.h>
2  #define LedPortSel  &P9SEL
3  #define LedPortDir  &P9DIR
4  #define LedPort     &P9OUT
5  Bit0 EQU 0x01 ; assign value (only value) to a symbol
6  ;===== PortsSetUp routine Definitions=====
7      MODULE PortSetUp
8      PUBLIC P9Config, P1Config
9      RSEG CODE
10
11 P9Config    bic.b    #0xff,LedPortSel
12            bis.b    #0xff,LedPortDir
13            clr.b    LedPort
14            ret
15
16 P1Config    bic.b    #Bit0,&P1SEL
17            bic.b    #Bit0,&P1DIR
18            bis.b    #Bit0,&P1IES
19            bis.b    #Bit0,&P1IE
20            bic.b    #Bit0,&P1IFG ; reset of interrupt flag
21            ret
22
23      ENDMOD
24  ;===== Delay - routine Definitions =====
25      MODULE Delay
26      PUBLIC Delay
27      RSEG CODE
28
29 Delay       mov.w    #065530,R15
30 L1         dec.w    R15
31           jnz      L1
32           ret
33
34      ENDMOD
35  ;===== Adder - routine Definitions =====
36      MODULE Adder
37      PUBLIC Adder
38      RSEG CODE
39
40 Adder       mov      2(SP),R4
41           mov      4(SP),R5
42           add      R4,R5
43           mov      R5,4(SP)
44           ret
45
46      ENDMOD
47  ;=====
48      END

```



## קובץ 3 - Interrupts.s43

```

1  #include <msp430xG46x.h>
2  ;-----
3  ;          MACRO Defenition
4  ;-----
5  ShowNum      MACRO      Num
6                mov.b     Num,&P9OUT
7                ENDM
8  ;-----
9  Clear_LEDs   MACRO
10               mov.b     0x00,&P9OUT
11               ENDM
12 ;-----
13
14               EXTERN     Delay
15 ;-----
16 ;          PORT1 Interrupt Service Rutine of SW1=P1.0
17 ;-----
18               MODULE     ISRs_Definitions
19               PUBLIC     PORT1_ISR
20               RSEG        CODE
21
22 PORT1_ISR     Clear_LEDs
23               call       #Delay           ;delay for Debounce
24               bic.b      #0x01,&P1IFG    ; manual clear of
25 Mainloop      mov.b      #0x01,R14
26 Subloop       ShowNum    R14
27
28               call       #Delay
29               bit.b       #0x01,&P1IFG
30               jnz        off
31               rla.b       R14
32               tst.b       R14
33               jz          Mainloop
34               jmp         Subloop
35
36 off           Clear_LEDs
37               bic.b       #0x01,&P1IFG
38               reti
39
40               ENDMOD
41 ;=====
42               END

```